F# Tutorial

C. H. Ting

We assume that a large portion of our reader are not familiar with Forth, and that we need a Forth system that is easy to use, easy to teach, easy to understand, and running on the most prevalent Windows operating system on personal computers.

F# is a simple Forth system designed to run efficiently under Windows XP.  It is base on an eForth kernel, with extensions to invoke Windows services.  It is a simple platform that an averaged user can understand and make use of the Windows operating system to develop his own programs.

F# is developed by eForth Technology, Inc., a company based in Hsinchu, Taiwan.  It is donated to the public domain to encourage people to learn Forth and use it to build applications on PC’s running Windows operating system.  The system portion of F# was written by Mr. C. S. Yap, and much of the application code was written by Mr. Sam Chen.

1.
eForth 

Forth is a programming language with its own operating system.  It can be implemented on any reasonably capable CPU and computer.  It has very simple syntax and rules of usage, and can be mastered in very short period of time.  Using Forth, one can easily write and debug substantial programs with high quality and in short time.  With a few simple extensions, the Forth system can be used as the operating system of an embedded system to produce firmware for a wide range of engineering products.

A Forth system is a collection of commands residing in memory.  These commands can be executed interactively, and they are also used to build new commands interactively.  Building new commands to replace lists of existing commands is a process of abstraction, aligning the computer closer and closer to the actual application.  In the end, the commands at the highest level become the application itself.  Forth commands have many of the nice attributes of objects in object oriented programming, but without the complications and bulkiness.  It is ideally suited for applications in small microcontrollers.  It is the best tool for firmware engineering.

Advantages of using Forth in firmware engineering are:

· High level programming language

· Platform independence

· Tight object code

· Interactive testing and debugging

· Ideal programming environment

The hardware requirements to run the eForth efficiently are a CPU with a reasonably good architecture, 16K bytes of ROM/RAM memory, and a serial UART port.  Now a days, these are very modest requirements for most microcontroller based applications.

Most embedded systems do not need a serial port when running stand-alone.  However, a serial port is very handy when servicing the system, and extremely useful when developing and debugging a product.  The serial port gives the programmer an easy channel into the target system.  He can interactively access all the hardware resources in the target system.  It makes interactive testing and debugging possible, and greatly eases the program development process.  It also allows the system to communicate with a host computer on the automated factory floor.

The eForth is designed for an idealized microcontroller system configured as shown in Figure 1.  It has a CPU with some memory, perhaps with some custom peripherals for specific applications.  It is connected to the outside world through a RS232 port.  This RS232 port may not be used in the actual applications, but it can be connected to a host computer for programming, testing and debugging.  In certain applications, the RS232 umbilical cord may never be severed because then the host can download new code to the target and send commands to the target to perform specific tasks.  Many of the new generation of instruments and equipment are designed to work either stand-alone and with a host computer.

There are many trade-offs that we have to consider in designing this eForth, which we assume will be ported to many different CPU's.  Here is a list of important features we strife to include in the eForth:

· .A small set of primitive commands, which is machine dependent.  Only this set of commands have to be rewritten for a specific CPU.  A minimum primitive commands set encourages the porting of eForth to new CPU's.

· All high level commands are built on the primitive commands.  The high level commands can be easily ported without changes to all target CPU's, including 8, 16, and 32 bit microprocessors and microcontrollers.

· The only I/O commands are KEY, EMIT, and ?KEY, because the only required I/O device in the target is a RS232 port.

· .Source code can be downloaded from the host computer and compiled by the target.

· .Core image of the system including application can be uploaded to host for programming EPROM to run the target system.

· .Editor, file management, and software version control are provided by the host system.  We assume that the HyperTerminal in Windows will be the host interface to the target system.

Firmware engineering will be very easy if we are dealing with microcontrollers and embedded system which meet the above requirements and have a Forth operating system to support the application.  It is then possible to develop substantial applications interactively on the target system.  Programs can be written in high level languages, and downloaded to the target system for incremental compilation and testing.  Program modules can be tested exhaustively before using them to build the modules in the next higher level.  Products can be debugged easily and problems can be fixed quickly.  

In this Firmware Engineering Workshop, I will discuss in minute details of this eForth system, how to use it to write application programs, and how to integrate the application program into the Forth system to produce firmware which will run he target embedded system.

2.
The most commonly used eForth Commands

Here is a list of 54 Forth commands most frequently used in Forth programs.  They are listed in alphabetic order so you can find them quickly.  They are used in the exercises and the examples in the following two sections.  A more comprehensive list of eForth commands are in the Appendix.

Stack Comments:

Stack inputs and outputs are shown in the form:  ( input1 input2 ... -- output1 output2 ... ) 

Unless noted otherwise, all numbers are 16 bits.  

flag 




Boolean flag, either 0 or -1




?




Truth value, either 0 or non-zero

char




ASCII character





b 




Any 8 bit number

x




Any 16 bit number




n




Signed 16 bit number 

u




Unsigned 16 bit number 



w




Wrap-around/circular 16 bit number

+n 




Positive 16 bit number 




addr 




Address (same as u, unsigned) 

d




Signed double (32 bit) number



xd, ud, wd, +d



Specify types of 32 bit number

!      

(x addr -- )          

Store x at addr.

( comment) 
( -- )            

Ignore comment text. 

*      

(w1 w2 -- w3)         
Multiply. "times"

*/     

(n1 n2 n3 -- n4)      
Leave quotient of (n1*n2)/n3. 

*/MOD

(n1 n2 n3 – rem quot)
Leave remainder and quotient of (n1*n2)/n3

+      

(w1 w2 -- w3) 

Add w1 and w2. 

+!     

(w addr -- )          

Add w to number at addr. 

-      

(w1 w2 -- w3) 

Subtract w2 from w1 (w1-w2=w3). 

.      

(n -- )               

Display signed number with trailing blank.

." <text>" 
( -- )              

Compile <text> message. At run-time display text message till “.

.R

(n1 n2 -- )

Display n1 right justified in n2 columns


.S

( -- )


Display the contents of the data stack.

/      

(n1 n2 -- quot)       
Division, signed (quotient of n1/n2). 

2DROP

(n n -- )


Discard two numbers.

2DUP

(n1 n2 – n1 n2 n1 n2)
Duplicate two numbers.

: <name> 
( -- )              

Begin a new command of <name>. 

;      

( -- )                

Terminate a new  command. 

<      

(n1 n2 -- ?) 

True if n1 less than n2. 

=     

 (n1 n2 -- ?) 
      
True if n1 equals n2. 

@      

(addr -- x)           

Replace addr by number at addr. 

BASE   

( -- addr ) 

Contain radix for input-output conversion. 

BEGIN

( -- )


Start an indefinite loop.

BL

( -- 32)


Put 32 on top of stack.

CELLS

(n – 2n)


Multiple n by 2.

CR     

( -- )                

Display a new line. 

DECIMAL 
( -- )               

Set number base to decimal.

DROP   

(x -- )


Discard top of stack.

DUP    

(x1 -- x2)

Duplicate top of stack.

ELSE

( -- )       

Terminate <true> clause, continue after the THEN.

EMIT   

( char -- )           

Display char.  

EXIT   

( -- )                

Terminate execution of a colon command.

FOR 

( limit -- )  

Setup loop. Repeat loop until limit is decremented to 0.

HEX

( -- )


Set number base to 16.

IF

( ? -- )     

If ? is zero, branches forward to <false> or after THEN. 

KEY    

( -- char )           

Get an ASCII character from the keyboard.  Does not echo.

MOD    

(n1 n2 -- mod)        
Modulus, signed (remainder of n1/n2).

NEGATE 
(n1 -- n2)            

Two's complement.

NEXT

( -- )     


Decrement index and repeat loop until index is less than 0

NUMBER?
(addr – n -1 | addr 0 )
Convert string at addr to a number n and a -1.  

Return addr and 0 if failed.

OR     

(x1 x2 -- x3)         
Logical bit-wise OR.

OVER   

(x1 x2 -- x1 x2 x1) 
Make copy of second item on stack.

QUERY

( -- )


Wait for one line of text from the terminal.

R@      

( -- w )              

Used inside a loop to get the current index value. 

ROT

(x1 x2 x3 -- x2 x3 x1)
Rotate third item to top. "rote"

SPACE  
( -- )                

Display a space.

SPACES 
( +n -- )             

Display +n spaces. 

SWAP 

(x1 x2 -- x2 x1)

Exchange top two stack items.

THEN

( -- )


Terminate the IF-ELSE structure.

U.R

(u n -- )


Display unsigned number u in n columns.

U<

(u1 u2 – ? )

Return true if u1<u2.

UM*

(u1 u2 – ud)

Multiply unsigned numbers. Return unsigned double product

UNTIL

( ? -- )     

Repeat <loop-body> until the ? flag is non-zero.

VARIABLE <name>
( -- )       
Define a variable.  At run-time, <name> leaves its address.

WORD <text>
(char -- addr )

Get the char delimited string <text>  from the input stream

3.
Install F# System
To install F# system on your computer, do the following:

Copy F#221.ZIP to a working folder.

Unzip all the files to a F# folder, for example D:\F#\

In Windows Manager, navigate to D:F# folder and double click F#.EXE.  You will see a file selection box, listing a number of .FEX files:


CONSOLE.FEX

DEMO.FEX


EXTEND.FEX


HELLO.FEX


SERIAL.FEX

Select one of these files, and you will run one of the demo programs:

HELLO.FEX
Show a small dialog box containing the classical greeting:

“Hello, World! from F#.”
DEMO.FEX
Open two windows, a graphic window showing flags and some graphic images, and a console window to sequence through the images on input keys of “B” (for previous image), “N” (for next image) and ESC (exit).

CONSOLE.FEX
Open a Forth console window to accept Forth commands.

EXTEND.FEX
Open a Forth console window, and load in many useful tools for writing and tests substantial Forth applications.  The tools include:


ANSI compatible word set


Assembler


Decompiler


Single step debugger

You should leave the files in D:\F# as there are and not disturb them, to make sure that you will always have a working F# system in this folder.  For your applications, you should open their own folders and use the .FEX file model to access F# system in D:\F# folder.

For example, create a new folder D:\APPLICATIONS\TEST1 and copy CONSOLE.FEX file into this folder.  Use your favorite text editor to edit this file.  Make a global replacement to replace “FLOAD “ with “FLOAD D:\F#\” so that all the files referred are properly redirected to the original files in D:\F#\ folder.

Do the same replacements in CONSOLE.F file, which loads in CONMENU.F and BUFFERIO.F.  You can add new FLOAD commands at the end of CONSOLE.FEX to load in your application programs.

Now, right click CONSOLE.FEX will open a selection box.  Select new program option to open a Program Selection Window.  Select Browse option here and navigate to F#.EXE in D:\F#\ and associate F#.EXE as the execution program of CONSOLE.FEX.

After associating F#.EXE with CONSOLE.FEX, you can invoke F# and load all the files listed in CONSOLE.FEX to bring up your application in D:\APPLICATIONS\TEST1\.  Similar procedure will allow you to place your application programs anywhere in your hard disk and invoke F# locally.

4.
Exercises and Examples 

There are 17 lessons to help you get familiar with Forth language, and show you how to write programs in Forth.  It is very important that you type in a few of these examples and make them work.  You will encounter many problems in entering the examples, like miss-typing characters, missing the spaces, entering wrong numbers, and stack underflows.  However, this is what it takes to learn a new language.  Of the 220 some eForth commands, only 54 commands are used in these examples.  Knowing how these 54 commands work will allow you to write Forth programs to solve most programming problems.
4.1
The Universal Greeting 

This is the universal greeting, the first programming example in all programming language manuals.  It requires 4 Forth commands and their formal descriptions are:

: <name> 
( -- )              

Begin a new command of <name>. 

;      

( -- )                

Terminate a new  command. 

 CR     

( -- )                

Display a new line. 

." <text>" 
( -- )              

Compile <text> message. At run-time display text message till “.

We build a new command HELLO this way:

: HELLO CR ." Hello, world!" ;

Type HELLO with a <return> displays the message ello, world!.’

4.2
The Big F 

The command .” can be used to build many other commands which display patterns on the terminal.  Two such commands are needed to display a big F character on the terminal:

: bar   CR ." *****" ;

: post  CR ." *    " ;

: F     bar post bar post post post ;

Type 'F' and a return on your keyboard, and you will see a large F character displayed on the screen 

4.3
Example 3.      FIG, Forth Interest Group 

A few more patterns will allow us to display any big ASCII characters on the terminal.

: center CR ."   *  " ;

: sides  CR ." *   *" ;

: triad1 CR ." * * *" ;

: triad2 CR ." **  *" ;

: triad3 CR ." *  **" ;

: triad4 CR ."  *** " ;

: quart  CR ." ** **" ;

: right  CR ." * ***" ;

: bigT  bar center center center center center center ;

: bigI  center center center center center center center ;

: bigN  sides triad2 triad2 triad1 triad3 triad2 sides ;

: bigG  triad4 sides post right triad1 sides triad4 ;

: FIG  F bigI bigG ;

4.4
Repeated Patterns 

We will use the .” command to display blocks of asterisks in various forms and shapes.  What we need is the looping mechanism to display a sequence of asterisks.  Here are these Forth commands needed:

FOR 

( limit -- )  

Setup loop. Repeat loop until limit is decremented to 0.

NEXT

( -- )     


Decrement index and repeat loop until index is less than 0

R@      

( -- w )              

Used inside a loop to get the current index value. 

VARIABLE <name>
( -- )       
Define a variable.  At run-time, <name> leaves its address.

@      

(addr -- x)           

Replace addr by number at addr. 

!      

(x addr -- )          

Store x at addr.

Variables points to locations in memory where data can be stored and retrieved.  @ retrieves data and ! stores data.

VARIABLE WIDTH                  ( number of asterisks to print )

: ASTERISKS ( -- , print n asterisks on the screen, n=width )

        WIDTH @                 ( limit=width, initial index=0 )

        FOR ." *"               ( print one asterisk at a time )

        NEXT                    ( repeat n times )

        ;

: RECTANGLE ( height width -- , print a rectangle of asterisks )

        WIDTH !                 ( initialize width to be printed )

        FOR     CR

                ASTERISKS       ( print a line of asterisks )

        NEXT

        ;

: PARALLELOGRAM ( height width -- )

        WIDTH !

        FOR     CR R@ SPACES    ( shift the lines to the right )

                ASTERISKS       ( print one line )

        NEXT

        ;

: TRIANGLE ( width -- , print a triangle area with asterisks )

        FOR     CR

                R@ WIDTH !      ( increase width every line )

                ASTERISKS       ( print one line )

        NEXT

        ;

Try the following instructions:

        3 10 RECTANGLE

        5 18 PARALLELOGRAM

        12 TRIANGLE  

4.5
The Theory That Jack Built 

This example shows you how to build a hierarchical structure in Forth.  From simple phrases, we can construct verses of a recursive poem.  Here is a few new commands we need to build the hierarchy.

IF

( ? -- )     

If ? is zero, branches forward to <false> or after THEN. 

ELSE

( -- )       

Terminate <true> clause, continue after the THEN.

THEN

( -- )


Terminate the IF-ELSE structure.

KEY    

( -- char )           

Get an ASCII character from the keyboard.  Does not echo.

SPACE  
( -- )                

Display a space.

SPACES 
( +n -- )             

Display +n spaces. 

DROP   

(x -- )


Discard top of stack.

: the           ." the " ;

: that          CR ." That " ;

: this          CR ." This is " the ;

: jack          ." Jack Builds" ;

: summary       ." Summary" ;

: flaw          ." Flaw" ;

: mummery       ." Mummery" ;

: k             ." Constant K" ;

: haze          ." Krudite Verbal Haze" ;

: phrase        ." Turn of a Plausible Phrase" ;

: bluff         ." Chaotic Confusion and Bluff" ;

: stuff         ." Cybernatics and Stuff" ;

: theory        ." Theory " jack ;

: button        ." Button to Start the Machine" ;

: child         ." Space Child with Brow Serene" ;

: cybernatics   ." Cybernatics and Stuff" ;

: hiding        CR ." Hiding " the flaw ;

: lay           that ." Lay in " the theory ;

: based         CR ." Based on " the mummery ;

: saved         that ." Saved " the summary ;

: cloak         CR ." Cloaking " k ;

: thick         IF that ELSE CR ." And " THEN

                ." Thickened " the haze ;

: hung          that ." Hung on " the phrase ;

: cover         IF that ." Covered "

                ELSE CR ." To Cover "

                THEN bluff ;

: make          CR ." To Make with " the cybernatics ;

: pushed        CR ." Who Pushed " button ;

: without       CR ." Without Confusion, Exposing the Bluff" ;

: rest                                  ( pause for user interaction )

        ." . "                          ( print a period )

        10 SPACES                       ( followed by 10 spaces )

        KEY                             ( wait the user to press a key )

        DROP CR CR CR ;

: cloaked cloak saved based hiding lay rest ;

: THEORY

        CR this theory rest

        this flaw lay rest

        this mummery hiding lay rest

        this summary based hiding lay rest

        this k saved based hiding lay rest

        this haze cloaked

        this bluff hung 1 thick cloaked

        this stuff 1 cover hung 0 thick cloaked

        this button make 0 cover hung 0 thick cloaked

        this child pushed

                CR ." That Made with " cybernatics without hung

                CR ." And, Shredding " the haze cloak

                CR ." Wrecked " the summary based hiding

                CR ." And Demolished " the theory rest

        ;

Type THEORY to start.  Hit any key to continue.

4.6
Help 

This example show you how to use Forth interpreter to carry on a dialog.  It pretends to be a psychiatrist talking to you as his patient.  You can add more topics for discussion if you like.

New commands appearing here are:

=     

 (n1 n2 -- ?) 
      
True if n1 equals n2. 

OR     

(x1 x2 -- x3)         
Logical bit-wise OR.

: question

        CR CR ." Any more problems you want to solve?"

        CR ." What kind ( sex, job, money, health ) ?"

        CR

        ;

: help  CR

        CR ." Hello!  My name is Creating Computer."

        CR ." Hi there!"

        CR ." Are you enjoying yourself here?"

        KEY 32 OR 89 =

        CR

        IF      CR ." I am glad to hear that."

        ELSE    CR ." I am sorry about that."

                CR ." maybe we can brighten your visit a bit."

        THEN

        CR ." Say!"

        CR ." I can solved all kinds of problems except those dealing"

        CR ." with Greece. "

        question

        ;

: sex   CR CR ." Is your problem TOO MUCH or TOO LITTLE?"

        CR

        ;

: too  ;                                ( noop for syntax smoothness )

: much  CR CR ." You call that a problem?!!  I SHOULD have that problem."

        CR ." If it reall y bothers you, take a cold shower."

        question

        ;

: little

        CR CR ." Why are you here!"

        CR ." You should be in Tokyo or New York of Amsterdam or"

        CR ." some place with some action."

        question

        ;

: health

        CR CR ." My advise to you is:"

        CR ."      1. Take two tablets of aspirin."

        CR ."      2. Drink plenty of fluids."

        CR ."      3. Go to bed (along) ."

        question

        ;

: job   CR CR ." I can sympathize with you."

        CR ." I have to work very long every day with no pay."

        CR ." My advise to you, is to open a rental computer store."

        question

        ;

: money

        CR CR ." Sorry!  I am broke too."

        CR ." Why don't you sell encyclopedias of marry"

        CR ." someone rich or stop eating, so you won't "

        CR ." need so much money?"

        question

        ;

: HELP help ;

: H help ;

: h help ;

Most of the commands defined in this exercise as in lower case.  Release the Caps Lock on your keyboard before proceeding.  Type 'help' to start.

4.7
Money Exchange 

This is the first example we will use to demonstrate how numbers are used in Forth.  It is a money exchange program, which converts money represented in different currencies.  Let's start with the following currency exchange table:

        33.55 NT        
1 Dollar

        7.73 HK         
1 Dollar

        9.47 RMB        
1 Dollar

        1 Ounce Gold    
285 Dollars

        1 Ounce Silver  
4.95 Dollars 

Numbers entered from the terminal are pushed on the first-in-last-out stack.  Commands get their input numbers from the stack and put the results back on the stack.  You have to be keenly aware of the numbers on the stack to be successful in programming Forth.

+      

(w1 w2 -- w3) 

Add w1 and w2. 

-      

(w1 w2 -- w3) 

Subtract w2 from w1 (w1-w2=w3). 

*      

(w1 w2 -- w3)         
Multiply. "times"

/      

(n1 n2 -- quot)       
Division, signed (quotient of n1/n2). 

*/     

(n1 n2 n3 -- n4)      
Leave quotient of (n1*n2)/n3. 

.      

(n -- )               

Display signed number with trailing blank.

.S

( -- )


Display the contents of the data stack.

The command */ is a scaling command, which converts quantities from one unit to another.  Choosing the scaling factors properly, you can use integers to do most of the things that require floating point math operations.  The command .S displays all the numbers on the stack.  It is very useful to inspect the stack at any time to make sure that you have the right numbers on the stack.

: NT    ( nNT -- $ )    100 3355 */  ;

: $NT   ( $ -- nNT )    3355 100 */  ;

: RMB   ( nRMB -- $ )   100 947 */  ;

: $RMB  ( $ -- nJmp )   947 100 */  ;

: HK    ( nHK -- $ )    100 773 */  ;

: $HK   ( $ -- $ )      773 100 */  ;

: GOLD  ( nOunce -- $ ) 285 *  ;

: $GOLD ( $ -- nOunce ) 285 /  ;

: SILVER ( nOunce -- $ ) 495 100 */  ;

: $SILVER ( $ -- nOunce ) 100 495 */  ;

: OUNCE ( n -- n, a word to improve syntax )  ;

: DOLLARS ( n -- )      . ;

With this set of money exchange words, we can do some tests:

        5 ounce gold .

        10 ounce silver .

        100 $NT .

        20 $RMB .

If you have many different currency bills in your wallet, you can add then all up in dollars:

        1000 NT 500 HK + .S

        320 RMB + .S

        DOLLARS ( print out total worth in dollars )

4.8
Temperature Conversion

Converting temperature readings between Celsius and Fahrenheit is also an interesting problem.  The difference between temperature conversion and money exchange is that the two temperature scales have an offset besides the scaling factor. 

: F>C ( nFahrenheit -- nCelsius )

        32 -

        10 18 */

        ;

: C>F ( nCelsius -- nFahrenheit )

        18 10 */

        32 +

        ;

Try these commands

90 F>C .        shows the temperature in a hot summer day and

0 C>F .         shows the temperature in a cold winter night. 

4.9
Weather Reporting. 

This is a good place to introduce the number comparing commands:

<      

(n1 n2 -- ? ) 

True if n1 less than n2. 

=     

(n1 n2 -- ? ) 
      
True if n1 equals n2. 

U<

(u1 u2 – ? )

Return true if u1<u2.

These commands compare the top two numbers on the stack and generate a flag, which is used by IF to determine which of the two alternate branches to execute.

: WEATHER ( nFarenheit -- )

        DUP     55 <

        IF      ." Too cold!" DROP

        ELSE    85 <

                IF      ." About right."

                ELSE    ." Too hot!"

                THEN

        THEN

        ;

You can type the following instructions and get some responses from the computer:

        90 WEATHER Too hot!

        70 WEATHER About right.

        32 WEATHER Too cold.

4.10
Print the multiplication table 

This exercise prints the multiplication table nicely formatted in columns and rows.  These are the commands necessary to format numbers in regular columns:

.R

(n1 n2 -- )

Display n1 right justified in n2 columns


U.R

(u n -- )


Display unsigned number u in n columns.

SPACE  
( -- )                

Display a space.

SPACES 
( +n -- )             

Display +n spaces. 

: ONEROW ( nRow -- )

        CR

        DUP 3 .R 3 SPACES

        1 11

        FOR     2DUP *

                4 .R

                1 +

        NEXT

        DROP ;

: MULTIPLY ( -- )

        CR CR 6 SPACES

        1 11

        FOR     DUP 4 .R 1 +

        NEXT DROP 

        1 11

        FOR     DUP ONEROW 1 +

        NEXT DROP

        ;

Type MULTIPLY to print the multiplication table 

4.11
Calendar

This exercise allows you to print weekly calendars for any month in any year, nicely formatted on the terminal.  It assumes that there are 1461 days in a 4 year period, 365 days in a year plus one leap day.  It starts from 1/1/1950 and is good to 2050 AD.

New commands used here are:

*/MOD

(n1 n2 n3 – rem quot)
Leave remainder and quotient of (n1*n2)/n3

+!     

(w addr -- )          

Add w to number at addr. 

EMIT   

( char -- )           

Display char.  

VARIABLE JULIAN                         ( 0 is 1/1/1950, good until 2050 )

VARIABLE LEAP                           ( 1 for a leap year, 0 otherwise. )

: YEAR ( YEAR --, compute Julian date and leap year )

        DUP

        1949 - 1461 4 */MOD             ( days since 1/1/1949 )

        365 - JULIAN !                  ( 0 for 1/1/1950 )

        3 =                             ( modulus 3 for a leap year )

        IF 1 ELSE 0 THEN                ( leap year )

        LEAP !

        DUP 2000 =                      ( 2000 is not a leap year )

        IF 0 LEAP ! THEN

        2001 <                          ( correction due to 2000 )

        IF ELSE -1 JULIAN +! THEN

        ;

: FIRST ( MONTH -- 1ST, 1st of a month from Jan. 1 )

        DUP 1 =

        IF DROP 0 EXIT THEN             ( 0 for Jan. 1 )

        DUP 2 =

        IF DROP 31 EXIT THEN            ( 31 for Feb. 1 )

        DUP 3 =

        IF DROP 59 LEAP @ + EXIT THEN   ( 59/60 for Mar. 1 )

        4 - 30624 1000 */

        90 + LEAP @ +                   ( Apr. 1 to Dec. 1 )

        ;

: STARS 60 FOR 42 EMIT NEXT ;           ( form the boarder )

: HEADER ( -- )                         ( print title bar )

        CR STARS CR 

        ."      SUN     MON     TUE     WED     THU     FRI     SAT"

        CR STARS CR                     ( print weekdays )

        ;

: BLANKS ( MONTH -- )                   ( skip days not in this month )

        FIRST JULIAN @ +                ( Julian date of 1st of month )

        7 MOD 8 * SPACES ;              ( skip colums if not Sunday   )

: DAYS ( MONTH -- )                     ( print days in a month )

        DUP FIRST                       ( days of 1st this month )

        SWAP 1 + FIRST                  ( days of 1st next month )

        OVER - 1 -                      ( loop to print the days )

        1 SWAP                          ( first day count -- )

        FOR  2DUP + 1 -

                JULIAN @ + 7 MOD        ( which day in the week? )

                IF ELSE CR THEN         ( start a new line if Sunday )

                DUP  8 U.R              ( print day in 8 column field )

                1 +

        NEXT

        2DROP ;                         ( discard 1st day in this month )

: MONTH ( N -- )                        ( print a month calendar )

        HEADER DUP BLANKS               ( print header )

        DAYS CR STARS CR ;              ( print days   )

: JANUARY       YEAR 1 MONTH ;

: FEBRUARY      YEAR 2 MONTH ;

: MARCH         YEAR 3 MONTH ;

: APRIL         YEAR 4 MONTH ;

: MAY           YEAR 5 MONTH ;

: JUNE          YEAR 6 MONTH ;

: JULY          YEAR 7 MONTH ;

: AUGUST        YEAR 8 MONTH ;

: SEPTEMBER     YEAR 9 MONTH ;

: OCTOBER       YEAR 10 MONTH ;

: NOVEMBER      YEAR 11 MONTH ;

: DECEMBER      YEAR 12 MONTH ;

To print the calendar of April 1999, type:

        1999 APRIL

4.12
Sine and Cosine

Sine and cosine of angles are among the most often encountered transdential functions, useful in drawing circles and many other different applications.  They are usually computed using floating numbers for accuracy and dynamic range.  However, for graphics applications in digital systems, single integers in the range from -32768 to 32767 are sufficient for most purposes.  We shall study the computation of sine and cosine using the single integers.

The value of sine or cosine of an angle lies between -1.0 and +1.0.  We choose to use the integer 10000 in decimal to represent 1.0 in the computation so that the sine and cosine can be represented with enough precision for most applications.  Pi is therefore 31416, and 90 degree angle is represented by 15708.  Angles are first reduced in to the range from -90 to +90 degrees, and then converted to radians in the ranges from -15708 to +15708.  From the radians we compute the values of sine and cosine.

The sine and cosine thus computed are accurate to 1 part in 10000.  This algorithm was first published by John Bumgarner in Forth Dimensions, Volume IV, No. 1, p. 7.

VARIABLE XS                             ( square of scaled angle )

: KN ( n1 n2 -- n3, n3=10000-n1*x*x/n2 where x is the angle )

        XS @ SWAP /                     ( x*x/n2 )

        NEGATE 10000 */                   ( -n1*x*x/n2 )

        10000 +                           ( 10000-n1*x*x/n2 )

        ;

: (SIN) ( x -- sine*10K, x in radian*10K )

        DUP DUP 10000 */                  ( x*x scaled by 10K )

        XS !                            ( save it in XS )

        10000 72 KN                       ( last term )

        42 KN 20 KN 6 KN                ( terms 3, 2, and 1 )

        10000 */                          ( times x )

        ;

: (COS) ( x -- cosine*10K, x in radian*10K )

        DUP 10000 */ XS !                 ( compute and save x*x )

        10000 56 KN 30 KN 12 KN 2 KN      ( serial expansion )

        ;

: SIN ( degree -- sine*10K )

        31415 180 */                       ( convert to radian )

        (SIN)                           ( compute sine )

        ;

: COS ( degree -- cosine*10K )

        31415 180 */

        (COS)

        ;

To test the routines, type:

        90 SIN .                         9999 

        45 SIN .                         7070 

        30 SIN .                         5000 

         0 SIN .                            0 

        90 COS .                            0 

        45 COS .                         7071 

         0 COS .                         10000

4.13
Square Root

There are many ways to take the square root of an integer.  The special routine here was first discovered by Wil Baden.  Wil used this routine as a programming challenge while attending a FORML Conference in Taiwan, 1984.

This algorithm is based on the fact that the square of n+1 is equal to the sum of the square of n plus 2n+1.  You start with an 0 on the stack and add to it 1, 3, 5, 7, etc., until the sum is greater than the integer you wished to take the root.  That number when you stopped is the square root.

: SQRT ( n -- root )

        65025 OVER U<                   ( largest square it can handle)

        IF DROP 255 EXIT THEN           ( safety exit )

        >R                              ( save square )

        1 1                             ( initial square and root )

        BEGIN                           ( set n1 as the limit )

                OVER R@ U<              ( next square )

        WHILE

                DUP CELLS 1 +           ( n*n+2n+1 )

                ROT + SWAP

                1 +                     ( n+1 )

        REPEAT

        SWAP DROP

        R> DROP

        ;

4.14
Radix for Number Conversions 

The computer handles numbers in binary.  For human consumption, numbers are converted to numeric strings for displaying.  The computer also accepts numbers in numeric strings entered by the user.  Radix and number base are problems humans invented for their own headaches.  Forth solves the problems simply by using a system variable BASE which controls the input and output number conversions.  We have this set of commands for number conversions:

BASE   

( -- addr ) 

Contain radix for input-output conversion. 

DECIMAL 
( -- )               

Set number base to decimal.

HEX

( -- )


Set number base to 16.

DECIMAL and HEX simply store 10 and 16, respectively, into BASE:

: DECIMAL 
10 BASE ! ;

: HEX

16 BASE ! ;

You can define commands to change number base to octal or binary: 

: OCTAL
8 BASE !  ;

: BINARY
2 BASE !  ;

Try converting numbers among different radices:

        DECIMAL 12345 HEX U.

        HEX ABCD DECIMAL U.

        DECIMAL 100 BINARY U.

        BINARY 101010101010 DECIMAL U.

Real programmers impress on novices by carrying a HP calculator which can convert numbers between decimal and hexadecimal.  A Forth computer has this calculator built in, besides other functions.

4.15
ASCII Character Table 

: CHARACTER ( n -- )

        DUP EMIT HEX DUP 3 .R

        OCTAL DUP 4 .R

        DECIMAL 3 .R

        2 SPACES

        ;

: LINE ( n -- )

        CR

        5 FOR   DUP CHARACTER

                16 +

        NEXT

        DROP ;

: TABLE ( -- )

        32

        15 FOR  DUP LINE

                1 +

        NEXT

        DROP ;

Type TABLE to display the ASCII character table.

4.16
Random Numbers

Random numbers are often used in computer simulations and computer games.  This random number generator was ublished in Leo Brodie's 'Starting Forth'.

VARIABLE RND                            ( seed )

HERE RND !                              ( initialize seed )

: RANDOM ( -- n, a random number within 0 to 65536 )

        RND @ 31421 *                   ( RND*31421 )

        6927 +                          ( RND*31421+6926, mod 65536)

        DUP RND !                       ( refresh he seed )

        ;

: CHOOSE ( n1 -- n2, a random number within 0 to n1 )

        RANDOM UM*                      ( n1*random to a double product)

        SWAP DROP                       ( discard lower part )

        ;                                ( in fact divide by 65536 )

To test the routine, type

        100 CHOOSE .

        100 CHOOSE .

        100 CHOOSE .

and verify that the results are randomly distributed between 0 and 99 . 

4.17
Guess a Number 

Forth is an operating system, besides being a programming language.  The Forth OS has its syntax rules in accepting commands and executing them.  An application, like this uess a Number’ game, often requires its own operating system with its own syntax rules to interact with the user.  This example shows you how to construct such an operating system, though in a very simple form.

The computer first asks the user to enter a number as the limit of the number to guess.  It than asks the user to enter the number it chooses.  It will tell the user if the number he just entered is higher or lower than the number it chose, until the user enter the right number.  Then it repeats the whole sequence.

As an operating system, the game requires lots more resources from Forth.  Here is the set of new commands that facilitates the programming of this game:

BEGIN

( -- )


Start an indefinite loop.

UNTIL

( ? -- )     

Repeat <loop-body> until the ? flag is non-zero.

QUERY

( -- )


Wait for one line of text from the terminal.

NUMBER?
(addr – n -1 | addr 0 )
Convert string at addr to a number n and a -1.  

Return addr and 0 if failed.

BL

( -- 32)


Put 32 on top of stack.

WORD <text>
(char -- addr )

Get the char delimited string <text>  from the input stream

2DROP

(n n -- )


Discard two numbers.

2DUP

(n1 n2 – n1 n2 n1 n2)
Duplicate two numbers.

The most important utility command is to ask the user to enter a number and push that number on the stack.  This is defined as a new command GetNumber as the following:

: GetNumber ( -- n )

        BEGIN

                CR ." Enter a Number: " ( show message )

                QUERY BL WORD NUMBER?   ( get a string )

        UNTIL                           ( repeat until a valid number )

        ;

With this utility command available, we can build this game 'Guess a Number.' 

: InitialNumber ( -- n , set up a number for the player to guess )

        CR CR CR ." What limit do you want?"

        GetNumber                       ( ask the user to enter a number )

        CR ." I have a number between 0 and " DUP .

        CR ." Now you try to guess what it is."

        CR

        CHOOSE                          ( choose a random number )

        ;                               ( between 0 and limit )

: Check ( n1 -- , allow player to guess, exit when the guess is correct )

        BEGIN   CR ." Please enter your guess."

                GetNumber

                2DUP =                  ( equal? )

                IF      2DROP           ( discard both numbers )

                        CR ." Correct!!!"

                        EXIT

                THEN

                OVER <

                IF      CR ." Too low."

                ELSE    CR ." Too high!"

                THEN    CR

        0 UNTIL                         ( always repeat )

        ;

: Greet ( -- )

        CR CR CR ." GUESS A NUMBER"

        CR ." This is a number guessing game.  I'll think"

        CR ." of a number between 0 and any limit you want."

        CR ." (It should be smaller than 32000.)"

        CR ." Then you have to guess what it is."

        ;

: GUESS ( -- , the game )

        Greet

        BEGIN   InitialNumber                   ( set initial number)

                Check                           ( let player guess )

                CR CR ." Do you want to play again? (Y/N) "

                KEY                             ( get one key )

                32 OR 110 =                     ( exit if it is N or n )

        UNTIL

        CR CR ." Thank you.  Have a good day."  ( sign off )

        CR

        ;

Type 'GUESS' will initialize the game and the computer will entertain a user for a while.  Note the use of the indefinite loop structure:

        BEGIN <repeat-clause> ( ? ) UNTIL

You can jump out of the infinite loop by the instruction EXIT, which skips all the instructions in a Forth command up to ';', which terminates this command and continues to the next command in the calling routine.

5.
Demo Programs

5.1.
The Simplest Windows Demo Program, HELLO.FEX

This program calls a Windows function MessageBoxA to display a short message in a small message box. The actual program consists of the following commands:

FLOAD winapi.f
LoadLibrary user32.dll 

4 WINAPI: MessageBoxA   
: hello ( -- ) 

  0 

  Z" Hello, World! from F#" 

  Z" Hello"   

  $40   

  MessageBoxA  

  DROP  

  BYE 

  ;  
hello   
HELLO.FEX contains very detailed comments on these commands.

5.2.
The Forth Programming Platform, EXTEND.FEX

This is the basic platform for most Forth applications on Windows.  It provides and console window for the user to enter commands and compile applications.  A set of most often used Windows functions are included, and more advanced Windows function can be added.

The file loading sequence is as following:

FLOAD init.f


initial stuff

FLOAD win32.f


win32 system interface

FLOAD consolei.f

api and constant defination

FLOAD ui.f


user interface helper routine ( reposition )

FLOAD console.f


the main console program

FLOAD conmenu.f

menu functions

FLOAD bufferio.f

buffered i/o function
FLOAD append.f


string functions

FLOAD extend.f 


load in utilities

EXTEND.F loads in another set of files containing many utility words.  The files and utility commands they bring in are:  
FLOAD MAP.F


CLink constants

FLOAD APPEND1.F

appendLast

FLOAD WAIT.F


break points

FLOAD DOLOOP.F

classic DO-LOOP structure

FLOAD CASE.F


CASE structure

FLOAD VALUE.F

 VALUE

FLOAD 486ASM.F

80486 assembler

fload redefine.f


words redefined in assembly

FLOAD ASCII.F


ASCII

FLOAD WORDS.F

a better WORDS

FLOAD sort.f


in-place sorting of n-cell array

fload log.f


log console output to a log-file

FLOAD SEE.F


a better SEE

FLOAD string.f


more string words

fload redefin9.f


redefine USER

FLOAD DBG.F


single-step debugger

On this fairly powerful platform, you can load in your application and test it.

5.3.
The Forth Console Program, CONSOLE.FEX

This is an advanced platform to explore Window soperating system and its capabilities.  Many utilities are added to the console window for the user to use by clicking on pull-down menus.  A set of most often used Forth functions are implemented as menu items, and more can be added if desirable.

The file loading sequence is as following:

FLOAD init.f


initial stuff

FLOAD win32.f


win32 system interface

FLOAD consolei.f

api and constant defination

FLOAD ui.f


user interface helper routine ( reposition )

FLOAD console.f


the main console program

FLOAD conmenu.f

menu functions

FLOAD bufferio.f

buffered i/o function
FLOAD constate.f 

status information

FLOAD append.f


string functions

FLOAD ansi.f


missing ANSI Forth words

FLOAD editori.f


Windows constants and functions for editor

FLOAD fileinc.f


Windows constants and functions for files

FLOAD editor.f


open window to edit a file

FLOAD textwin.f


window to display text

FLOAD wstackv.f

window to display data stack

FLOAD wdump.f 

window to dump memory

FLOAD wsee.f


window to decompile code

FLOAD wwords.f

window to display words

FLOAD extend.f


load in utilities

FLOAD ccg.f


load in applications

FLOAD winsock.f
A simple text editor is added as an menu item in the pull-down menu “Edit”, to open a separate editing window to manipulate text in a file.  The four classic Forth debugging words DUMP, WORDS, SEE and .S are implemented as menu items in the pull-down menu “Tools”.

Currently, only memory dump and stack dump are implemented.  It is left to the user to add the decompiler (SEE) and dictionary dump (WORDS) functions, as both WEE.F and WWORDS.F files are empty files.  Please refer to CONMENU.F to see how memory dump and stack dump are implemented as menu items, calling from the console window.

5.4.
Graphic Demonstration, DEMO.FEX

Windows have very extensive graphics capabilities, which you may want to take advantage of.  DEMO.FEX shows you how to make use of these capabilities.  It opens a separated graphics window called CANVAS for you to draw lines and shapes.  DEMO.FEX loads in the following files:

FLOAD init.f


initial stuff

FLOAD win32.f


win32 system interface

FLOAD consolei.f 

api and constant definition

FLOAD ui.f 


user interface helper routine

FLOAD console.f


open main console window

FLOAD canvasi.f


Windows constants and functions for graphics

FLOAD canvasE.f

open canvas window

FLOAD charsE.f


basic drawing commands

FLOAD ansi.f


ANSI Forth commands

FLOAD chinese.f


commands to draw flags and other shapes

FLOAD demo.f


loop to sequence through demo commands

CHINESE.F and DEMO.F is of special interests because all the programs in them are written in Chinese!  They serve two very important purposes in developing F# system originally: 

· To eliminate the language gap for non-English speaking users to master personal computers, especially for the billions of Chinese.

· To demonstrate that Forth can be taught to primary school kids in Taiwan and China before they learn English.

All Forth commands needed to construct the demo commands are redefined and given Chinese names.  Subsequently defined words are all given Chinese names.  These new commands are collected in two vocabularies, a primary vocabulary holding commands to draw simple shapes, and a secondary vocabulary with more advanced commands added.  For first time students, only the primary vocabulary is made available to them, so that they are not confused or intimidated by the huge Forth dictionary.  After they have mastered the command set in the primary vocabulary, they can open the secondary vocabulary to learn more about Forth and graphic operations.  For the daring students already familiar with English, the entire Forth dictionary is always available.

For Forth users in US and Europe, the commands redefined in CHINESE.F are all available with their original English names in CHARE.F file.

5.5.
An I/O Interface Example, SERIAL.FEX

SERIAL.FEX is an example on how to access I/O devices under Windows.  Windows do not allow user to read and write directly to peripheral devices.  One has to access the peripheral devices through Windows service calls.

SERIAL.FEX calls the following files to prepare F# to access COM1 port:

FLOAD init.f 


include initial stuff

FLOAD win32.f 


include API interface

FLOAD consolei.f 

include API functions and constants

FLOAD ui.f 


include user interface helper routine ( reposition )

FLOAD console.f 

include the procedure to open console
It then defines all the Windows API’s it needs and builds a test loop to read inputs from COM1 and writes output to COM1.  To run TEST, make sure that COM1 port is connected to an active RS232 device with the correct baud rate.  Or, jumper together pins 2 and 3 on the DB9 connector so that COM1 loops back on itself.

5.6.
Concluding Remarks

F# system is still under development.  However, the underlying platform has been quite stable for a few years now, and was used to develop several substantial applications.  It is hoped that once it is released into the public domain, more Forth programmers will use it to develop Windows applications, and more non-Forth programmers will have an opportunity to learn about this very unique language in solving practical programming problems.

Appendix
eForth Quick Reference
Stack Comments:

Stack inputs and outputs are shown in the form:  ( input1 input2 ... -- output1 output2 ... ) 

Unless noted otherwise, all numbers are 16 bits.  

flag 




Boolean flag, either 0 or -1




?




Truth value, either 0 or non-zero

char




ASCII character





b 




Any 8 bit number

x




Any 16 bit number




n




Signed 16 bit number 

u




Unsigned 16 bit number 



w




Wrap-around/circular 16 bit number

+n 




Positive 16 bit number 




addr 




Address (same as u, unsigned) 

d




Signed double (32 bit) number



xd, ud, wd, +d



Specify types of 32 bit number

Stack Manipulation Words

DUP    
(x1 -- x2)


Duplicate top of stack.

DROP   
(x -- )



Discard top of stack.

SWAP 
(x1 x2 -- x2 x1)


Exchange top two stack items.

OVER   
(x1 x2 -- x1 x2 x1) 

Make copy of second item on stack.

ROT
(x1 x2 x3 -- x2 x3 x1)

Rotate third item to top. "rote"

>R    
(x -- )  



Move top item to return stack for temporary storage. 

R>     
( -- x)



Retrieve top item from return stack. 

R@     
( -- x)               


Copy top of return stack onto stack.

Arithmetic Words

+      
(w1 w2 -- w3) 


Add w1 and w2. 

-      
(w1 w2 -- w3) 


Subtract w2 from w1 (w1-w2=w3). 

*      
(w1 w2 -- w3)         

Multiply. "times"

/      
(n1 n2 -- quot)       

Division, signed (quotient of n1/n2). 

MOD    
(n1 n2 -- mod)        

Modulus, signed (remainder of n1/n2).

*/     
(n1 n2 n3 -- n4)      

Leave quotient of (n1*n2)/n3. 

MAX    
(n1 n2 -- n3)         

n3 is the larger of n1 and n2.

MIN    
(n1 n2 -- n3)         

n3 is the smaller of n1 and n2.

ABS    
(n -- u)              


If n is negative, u is n's two's complement. 

NEGATE 
(n1 -- n2)            

Two's complement.

Logic and Comparison Words

NOT    
(x1 -- x2)            


Bit-wise one's complement.

AND    
(x1 x2 -- x3)         

Logical bit-wise AND.

OR     
(x1 x2 -- x3)         

Logical bit-wise OR.

XOR   
(x1 x2 -- x3)         

Logical bit-wise exclusive OR. 

<      
(n1 n2 -- flag) 


True if n1 less than n2. 

=     
 (n1 n2 -- flag)       

True if n1 equals n2. 

>      
(n1 n2 -- flag)       

True if n1 greater than n2. 

0<     
(n -- flag)           


rue if n is negative. 

Memory and Character String Words

@      
(addr -- x)           


Replace addr by number at addr. 

!      
(x addr -- )          


Store x at addr.

C@     
(addr -- b)           


Fetch least-significant byte only. 

C!     
(b addr -- )          


Store least-significant byte only. 

+!     
(w addr -- )          


Add w to number at addr. 

CMOVE  
(addr1 addr2 +n -- )  
Move n bytes starting at addr1 to memory starting at addr2.

COUNT  
(addr1 -- addr+1 b )
Move string count from memory onto stack. 

Numberic Conversion Words

BASE   
( -- addr ) 


Contain radix for input-output conversion. 

DECIMAL 
( -- )               

Set number base to decimal.

Terminal Input-Output Words

.      

(n -- )               

Display signed number with trailing blank.

U.     

(u -- )   


Display unsigned number with trailing blank. 

." <text>" 
( -- )              

Compile <text> message. At run-time display text message.

.( <text>) 
( -- )           

Display <text> from the input stream.

CR     

( -- )                

Display a new line. 

EMIT   

( char -- )           

Display char.  

TYPE   

( addr +n -- )        
Display a string of +n characters starting at address addr. 

SPACE  
( -- )                

Display a space.

SPACES 
( +n -- )             

Display +n spaces. 

KEY    

( -- char )           

Get an ASCII character from the keyboard.  Does not echo.

Compiler and Interpreter Words

( comment) 
( -- )            

Ignore comment text. 

,      

(x -- )               

Add x to parameter field of the most recently defined word. 

ALLOT  
(+n -- )               

Add +n bytes to parameter field of the most recently word.

LITERAL 
(x -- )              

Compile number x.  At  run-time, x is pushed on the stack.

[      

( -- )                

Switch from compilation to interpretation. "left-bracket" 

]      

( -- )                

Switch from interpretation to compilation. "right-bracket"

WORD <text>
(char -- addr )

Get the char delimited string <text>  from the input stream 






and leave as a counted string at addr. 

Defining Words
: <name> 
( -- )              

Begin a colon command of <name>. 

;      

( -- )                

Terminate execution of a colon  command. 

CREATE <name> 
( -- )         
Dictionary entry with no parameter field space reserved. 

'  <name> 

( -- addr)      
Find <name> and leave its address.

VARIABLE <name>
( -- )       
Define a variable.  At run-time, <name> leaves its address.

CONSTANT <name>
( x -- ) 

Define a constant.  At run-time, x is left on the stack.

Compiler Structure Words

EXIT   

( -- )                

Terminate execution of a colon command.

EXECUTE 
( addr -- )      

Execute the word command associated with addr.

QUIT 
  
( -- )                

Return to terminal, no stack change, no message.

ABORT 
( -- )                 

Return to terminal, clear stack, no message.

Conditional branch:
IF

( ? -- )     

If ? is zero, branches forward to <false> or after THEN. 

ELSE

( -- )       

Terminate <true> clause, continue after the THEN.

THEN

( -- )


Terminate the IF-ELSE structure.

Definite loop:

FOR 

( limit -- )  

Setup loop. Repeat loop until limit is decremented to 0.

NEXT

( -- )     


Decrement index and repeat loop until index is less than 0

R@      

( -- w )              

Used inside a loop to get the current index value. 

Indefinite loops:

BEGIN

( -- )


Start an indefinite loop.

UNTIL

( ? -- )     

Repeat <loop-body> until the ? flag is non-zero.

WHILE          
 ( ? -- )     

Repeat <loop-body> and <true> clause while the ? is non-zero.

REPEAT 
( -- )       

Unconditional backward branch to <loop> clause.

Dictionary and Vocabulary Words

HERE   

( -- addr )           

Address of next available dictionary location.

PAD    

( -- addr )           

Leave address of a scratch area of at least 84 bytes.

TIB    

( -- addr )           

Address of the text input buffer. 

' <name> 
( -- comp-addr ) 

Look up <name> in the dictionary. 

['] <name> 
( -- )            

Compile the word <name> in the input stream as an literal.   FORGET <name>
( -- )         


Delete <name>  and all words  added afterwards. 

Utility Words

.S

( -- )


Display the contents of the data stack.

WORDS 
( -- )


Display all words which contain the string <text>.

DUMP 

( addr  n -- )

Dump n bytes of memory starting from addr.

SEE <name>
( -- )


Decompile the word <name>.

READ

( addr  n -- )

Read text file from terminal into file buffer

LOAD

( addr  n -- )

Load source code in the file buffer.

SEND  

( addr  n -- )

Upload memory to host in Intel Hex format.

